Math 2250-3
Wednesday October 29
Solving DE's in Maple

These notes discuss material from sections 5.5 and 5.6 of the text, in the context of Maple. They are online at our Maple page, go to www.math.utah.edu/~korevaar/2250fall03/2250maple.html.

Part 1: checking homework problems on Maple

Here's a homework problem worked out in Maple, **#12 from section 5.5**. This problem was fairly painful to work by hand. For more information on the commands which are used, use the Help files.

```
> with(DEtools):  #DE command library
> f:=x->2-sin(x):  #inhomogeneous term
  deqtn:=diff(y(x),x,x,x)+ diff(y(x),x)=f(x):
     #the differential equation, #12 page 337.
> dsolve(deqtn,y(x));
```

$$y(x) = -\_C2\cos(x) + \_C1\sin(x) + \cos(x) + \frac{1}{2}x\sin(x) + 2x + \_C3$$

Discarding the pieces which solve the homogeneous equation, the simplest particular solution we see is

```
> yp:=x->1/2*x*sin(x)+2*x;
```

$$yp := x \to \frac{1}{2}x\sin(x) + 2x$$

The method of guessing as modified for problems when the guess for yp involves homogeneous equation solutions, would have had us to a guess of the form yp=Ax + x(Bsin(x)+Ccos(x)). The algebra in computing L(yp) was messy. We could check the algebra part of our work as well as the final answer:

```
> yp:=x->A*x+x*B*sin(x)+x*C*cos(x);
```

$$yp := x \to A\,x + x\,B\sin(x) + x\,C\cos(x)$$

```
> diff(yp(x),x,x,x)+diff(yp(x),x)=2-sin(x);
```

$$-2B\sin(x) - 2C\cos(x) + A = 2 - \sin(x)$$

Equating coefficients we see that -2B=-1, C=0, and A=2. This leads to yp=(1/2)xsin(x)+2x, as before.

How about #37?

```
> f:=x->1+x*exp(x):  #right hand side, then DE:
  deqtn:= diff(y(x),x,x,x) -2*diff(y(x),x,x)
       + diff(y(x),x) =f(x);
  ics:=y(0)=0, D(y)(0)=0, D(D(y))(0)=1:
        #initial conditions...see help files for syntax!
  ans37:=dsolve({deqtn,ics},y(x));
```

$$deqtn := \left(\frac{d^3}{dx^3}y(x)\right) - 2\left(\frac{d^2}{dx^2}y(x)\right) + \left(\frac{d}{dx}y(x)\right) = 1 + x\,\mathbf{e}^x$$

$$ans37 := y(x) = -4\,\mathbf{e}^x + 3\,x\,\mathbf{e}^x + \frac{1}{6}x^3\,\mathbf{e}^x - \frac{1}{2}x^2\,\mathbf{e}^x + x + 4$$

Part 2: Forced Oscillators:

Example 1 page 340

```
> with(plots):
  plot(5*cos(3*t)-5*cos(5*t),t=0..6*Pi,color=black,
  title='Figure 5.6.2 page 341');
```



Beating:

If the forcing frequency is not equal to the natural frequency the general solution to will be the superposition of two cos-sin terms, one corresponding to the particular solution with angular frequency w, and the other being the general solution to the homogeneous problem, with angular frequency w0. When w and w0 as well as the corresponding amplitudes are close, the system exhibits **beating**. Musicians tune their instruments using this phenomenon.

```
> restart:with(DEtools):with(plots):
     #I want to undefine m, x0 etc.
Warning, the name changecoords has been redefined
```

```
> deqtn1a:=diff(x(t),t,t) + w0^2*x(t) = (F0/m)*cos(w*t);
     #this is the case w not equal to w0,
     #I divided the model equation by m
```

$$deqtn1a := \left(\frac{d^2}{dt^2}\,\mathrm{x}(t)\right) + w0^2\,\mathrm{x}(t) = \frac{F0\cos(w\,t)}{m}$$

```
> dsolve(deqtn1a,x(t));
    #general solution, equal to particular
    #solution plus general homogeneous eqtn solution.
```

$$\mathrm{x}(t) = \sin(w0\,t)\,\_C2 + \cos(w0\,t)\,\_C1 + \frac{F0\cos(w\,t)}{m\,(w0^2 - w^2)}$$

```
> sol4:=dsolve({deqtn1a,x(0)=0,D(x)(0)=0},x(t));
    #a nice choice of initial conditions
```

$$sol4 := \mathrm{x}(t) = \frac{\cos(w0\,t)\,F0}{m\,(-w0^2 + w^2)} + \frac{F0\cos(w\,t)}{m\,(w0^2 - w^2)}$$

```
> sol5:=subs({w=45,w0=55,F0=50,m=.1},sol4);
     #Example 2 page 342
```

$$sol5 := x(t) = -0.5000000000 \cos(55\, t) + 0.5000000000 \cos(45\, t)$$

```
> actual:=plot(rhs(sol5),t=0..3, color=black):
  envel1:=plot(sin(5*t),t=0..3,color=black,linestyle=2):
  envel2:=plot(-sin(5*t),t=0..3,color=black,linestyle=2):
  display({actual, envel1,envel2},title='Figure 5.6.3 page 342');
```

Figure 5.6.3 page 342



Resonance in undamped, forced harmonic oscillators, when w=w0: You can use variation of parameters or undetermined coefficients to solve the forced oscillator, with no damping, in the case that the driving frequency w exactly equals the natural frequency w0. Or, you can use Maple

```
> deqtn1:=diff(x(t),t,t) + w0^2*x(t) = (F0/m)*cos(w0*t);
     #I have again written k/m = w0^2
```

$$deqtn1 := \left(\frac{d^2}{dt^2} x(t)\right) + w0^2\, x(t) = \frac{F0 \cos(w0\, t)}{m}$$

```
> sol1:=dsolve(deqtn1,x(t));
      #general solution
```

$$sol1 := x(t) = \sin(w0\, t)\, \_C2 + \cos(w0\, t)\, \_C1 + \frac{1}{2} \frac{F0\, (\cos(w0\, t) + \sin(w0\, t)\, w0\, t)}{m\, w0^2}$$

You see there the particular solution which we found, namely xp(t):

```
> xp:=t->(1/2)*F0*t*sin(w0*t)/(w0*m);
```

$$xp := t \rightarrow \frac{1}{2} \frac{F0\, t \sin(w0\, t)}{w0\, m}$$

You can see that xp(t) solves the initial value problem x0=v0=0, i.e. the system initially at rest.
For example, here's how to make the picture on page 343 of the text:

```
> F0:=100:w0:=50:F0:=100:m:=1:
>
> xp(t);
```

$$t \sin(50\, t)$$

```
> plot(xp(t),t=0..1.5,color=black,title='Figure 5.6.4 page 343');
```



Figure 5.6.4 page 343
1
0.5
0
–0.5
–1
–1.5
0.2
0.4
0.6
0.8
1
1.2
1.4
t